1. Constant Definition 

A variable shall be interpreted as a constant if the value of the expression (rhs) can be computed statically. Example: 

  reg  [15:0] r;

  initial begin

    r = 16’h000A;

    r2 = 3* r; 

  end

2. Modeling Read-Only Memories (ROM)

Asynchronous ROMs instances may be modeled as combinational logic using any of the following styles: 

1. case statement of address with assignment to vector output (section 5.6.1).  

2. initial statement of a memory array with values defined in Verilog code (section 5.6.2).   A ROM instance may be generated when the memory array is read from within a procedural block.  The ROM instance shall not be written in any other procedural block or continuous assignment. 

3. initial statement with memory initialized with $readmemb or $readmemh. ROM data shall be in a text file (section 5.6.3).  A ROM instance may be generated when the memory array is read from within a procedural block.  The ROM instance shall not be written in any other procedural block or continuous assignment.  

The rom_block attribute may 
be used to help guide the synthesis process.  This attribute shall precede the declaration of the ROM instance. 

5.6.1 ROM Using One-Dimensional Array with case Statement

The data values of a ROM shall be defined within a case statement.  All the values of the ROM shall be defined within the case statement.  The value assigned to each ROM address shall be a static expression that can be evaluated at compile time. 

Example:  

// Verilog-95 style 

module rom_case(z, a);
  output [3:0] z; // 4 wide 
  input  [2:0] a; // address- 8 deep memory  

  (* synthesis, rom_block = "ROM_CELL XYZ01" *)

  reg    [3:0] z;

  always @* begin // @(a)
    case (a)
      3'b000:  z = 4'b1011;
      3'b001:  z = 4'b0001;
      3'b100:  z = 4'b0011;
      3'b110:  z = 4'b0010;
      3'b111:  z = 4'b1110;
      default: z = 4'b0000;
    endcase
  end
endmodule // rom_case


// verilog-2001 ANSI  C Style
module rom_case(

     (* synthesis, rom_block = "ROM_CELLXYZ01" *)

     output reg [3:0] z,

     input wire [2:0] a); // address- 8 deep memory

     always@* begin // @(a)

       case (a)

           3'b000: z = 4'b1011;

           3'b001: z = 4'b0001;

           3'b100: z = 4'b0011;

           3'b110: z = 4'b0010;

           3'b111: z = 4'b1110;

           default: z =4'b0000;

        endcase

      end

endmodule // rom_case

Note: See section 6.1 for additional information on the definition of the synthesis attributes.  

5.6.2  ROM Using Two-Dimensional Array within Initial Block

The values of the ROM may be defined within a constant defined as a two-dimensional array.  A rom_block or logic_block attribute may be necessary to guide the synthesis style of the synthesis tool.   Uninitialized values shall have a don't care assignment, and may be assigned  a value of 1 or 0. 

Example: 

// Verilog-95 style
module rom_2dimarray_inital (z, a);
  output [3:0] z;
  input  [2:0] a; // address- 8 deep memory

   // declare a memory rom of 8 4-bit registers. The indices are 0 to 7
  (* synthesis, rom_block = "ROM_CELL XYZ01" *)

  reg    [3:0] rom[0:7];  

  

  initial begin
    rom[0] = 4'b1011;
    rom[1] = 4'b0001;
    rom[2] = 4'b0011;
    rom[3] = 4'b0010;
    rom[4] = 4'b1110;
    rom[5] = 4'b0111;
    rom[6] = 4'b0101;
    rom[7] = 4'b0100;
  end
 
  assign z = rom[a];   

endmodule

// verilog-2001 ANSI  C Style
module rom_2dimarray_inital (

     output reg [3:0] z,

     input wire [2:0] a); // address- 8 deep memory

   // declare a memory rom of 8 4-bit registers. The indices are 0 to 7
  (* synthesis, rom_block = "ROM_CELL XYZ01" *)

  reg    [3:0] rom[0:7];  
  initial begin
    rom[0] = 4'b1011;
    rom[1] = 4'b0001;
    rom[2] = 4'b0011;
    rom[3] = 4'b0010;
    rom[4] = 4'b1110;
    rom[5] = 4'b0111;
    rom[6] = 4'b0101;
    rom[7] = 4'b0100;
  end
 
  assign z = rom[a];   

endmodule

 Note: If combinational logic is desired, replace the 
(* synthesis, rom_block = "ROM_CELL XYZ01" *) attribute with the 
(* synthesis, logic_block *) attribute in the above example.

5.6.3 ROM Using Two-Dimensional Array with Data in Text File

The values of a ROM may be defined within a two-dimensional  array using the initial statement with the values defined in text files.  The system functions $readmemb or $readmemh shall be used to identify the file.  The format of the file shall be as defined in the $readmemb or $readmemh system functions.  The following restrictions shall apply: 

1. The ROM shall be defined as a two-dimensional array.

2. An attribute may be necessary to identify the ROM style 
(*synthesis, rom_block *) or (*synthesis, logic_block *).

3. Uninitialized values shall have a don't care assignment, and can be assigned by the synthesis tool a value of 1 or 0.

The syntax for memory load system tasks is: 

load_memory_tasks ::=

$readmemb ( " file_name ", memory_name [ , start_addr [ , finish_addr ] ] ) ;

| $readmemh ( " file_name ", memory_name [ , start_addr [ , finish_addr ] ] ) ;

The text file to be read shall contain only the following:

— White space (spaces, new lines, tabs, and form-feeds)

— Comments (both types of comment are allowed)

— Binary or hexadecimal numbers

The numbers shall have neither the length nor the base format specified. For $readmemb, each number shall be binary.  For $readmemh, the numbers shall be hexadecimal.  The unknown value (x or X), the high impedance value (z or Z), and the underscore (_) can be used in specifying a number as in a Verilog HDL source description.  The unknown value and the high impedance value shall be interpreted as don’t care, and may me modeled as either 0 or 1.   White space and/or comments shall be used to separate the numbers.

The term “address” refers to an index into the array that models the memory.  As the file is read, each number encountered is assigned to a successive word element of the memory.  Addressing is controlled both by specifying start and/or finish addresses in the system task invocation and by specifying addresses in the data file.

When addresses appear in the data file, the format is an “at” character (@) followed by a hexadecimal number as follows: @hh...h    Both uppercase and lowercase digits are allowed in the number.  No white space is allowed between the @ and the number.  As many address specifications as needed within the data file may be used.  When the system task encounters an address specification, it loads subsequent data starting at that memory address.  If no addressing information is specified within the system task, and no address specifications appear within the data file, then the default start address is the left-hand address given in the declaration of the memory.  Consecutive words are loaded until either the memory is full or the data file is completely read. If the start address is specified in the task without the finish address, then loading starts at the specified start address and continues towards the right-hand address given in the declaration of the memory.

If both start and finish addresses are specified as parameters to the task, then loading begins at the start address and continues toward the finish address, regardless of how the addresses are specified in the memory declaration.  When addressing information is specified both in the system task and in the data file, the addresses in the data file shall be within the address range specified by the system task parameters; otherwise, an error message is issued and the load operation is terminated. A warning message is issued if the number of data words in the file differs from the number of words in the range implied by the start through finish addresses.

All uninitialized memory locations shall be treated as don’t care, and may be implemented as 0 or 1. 

Example:

reg [7:0] mem[1:256];

Given this declaration, each of the following statements will load data into mem in a different manner:

initial $readmemh("mem.data", mem);

initial $readmemh("mem.data", mem, 16);

initial $readmemh("mem.data", mem, 128, 1);

The first statement will load up the memory at simulation time 0 starting at the memory address 1.  The second statement will begin loading at address 16 and continue on towards address 256.  For the third and final statement, loading will begin at address 128 and continue down towards address 1.  In the third case, when loading is complete, a final check is performed to ensure that exactly 128 numbers are contained in the file. If the check fails, a warning message is issued.

Example: 

// Verilog-95 style
// ROM module using two dimensional arrays with 
// memory defined in text file with $readmemb or $readmemh
// NOTE: This style can lead to simulation/synthesis mismatch
//       if the content of data file changes after synthesis
module rom_2dimarray_initial_readmem (z, a);
  output [3:0] z;
  input  [2:0] a; 

  // declares a memory rom of 8 4-bit registers.

  //The indices are 0 to 7

  (* synthesis, rom_block = "ROM_CELL XYZ01" *)

  reg    [3:0] rom[0:7]; 
  // NOTE:  To infer combinational logic instead of a ROM, use
  // (* synthesis, logic_block *)

 

  initial $readmemb("rom.data", rom); 

  assign z = rom[a];
endmodule

// Verilog-2001 style
// ROM module using two dimensional arrays with 
// memory defined in text file with $readmemb or $readmemh
// NOTE: This style can lead to simulation/synthesis mismatch
//       if the content of data file changes after synthesis
module rom_2dimarray_initial_readmem ( 

  output wire [3:0] z,
  input  wire [2:0] a); 

  // declares a memory rom of 8 4-bit registers.

  //The indices are 0 to 7

  (* synthesis, rom_block = "ROM_CELL XYZ01" *)

  reg    [3:0] rom[0:7]; 
  // NOTE:  To infer combinational logic instead of a ROM, use
  // (* synthesis, logic_block *)

 
  initial $readmemb("rom.data", rom); 

  assign z = rom[a];
endmodule

Example of content “rom.data” file: 

// file: /user/name/project/design/rom/rom.data

// date : Jan 08, 02

1011  // addr=0

1_000  // addr=1

000_0 //  addr=2

10_00 // addr=3

0010 // addr=4

0101 // addr=5 

1111 // addr=6

1001 // addr=7 

5.7 Modeling Random Access Memories (RAM)

Synchronous RAMs shall be modeled as a two-dimensional array.  The data entry into the RAM shall be modeled as a clocked block, and the asynchronous READ shall be modeled as either an assign statement or a combinational block.  

Two RAM style attributes may be used to help guide the synthesis process. The following restrictions shall apply: 

1. The RAM shall be defined as a two-dimensional array.

2. An attribute may be necessary to identify the RAM style 
(* synthesis, ram_block *) for RAM cells, or 
(* synthesis, reg_block *)  for registers.

Example: 

// Verilog-95 style
module ram_test(q, a, d, we, clk);
  output [7:0] q;
  input  [7:0] d;
  input  [6:0] a;
  input        clk, we;

  (* synthesis, ram_block *)
  reg    [7:0] mem [127:0];

  always @(posedge clk) if(we) mem[a] <= d;

  assign q = mem[a];
endmodule


// Verilog-2001 style

module ram_test( 

  output wire [7:0] q, 
  input  wire [7:0] d,
  input  wire [6:0] a, 
  input  wire  clk, we);

  (* synthesis, ram_block *)
  reg    [7:0] mem [127:0];

  always @(posedge clk) if(we) mem[a] <= d;

  assign q = mem[a];
endmodule


//Ram implemented with latches

// Verilog-95 style

module ramlatch (q, a, d, we);
  output [7:0] q;  // output
  input  [7:0] d;  // data input
  input  [6:0] a;  // address
  input        we; // clock and write enable 

  // memory 128 deep, 8 wide

  (* synthesis, ram_block *)
  reg    [7:0] mem [127:0]; 


  always @* if(we) mem[a] <= d; 
 
  assign q = mem[a];
endmodule 
   

//Ram implemented with latches

// Verilog-2001 style

module ramlatch (
  output wire [7:0] q,  // output
  input  wire [7:0] d,  // data input
  input  wire [6:0] a,  // address
  input  wire       we); // clock and write enable 

  // memory 128 deep, 8 wide

  (* synthesis, ram_block *)
  reg    [7:0] mem [127:0];  


  always @* if(we) mem[a] <= d; 
 
  assign q = mem[a];
endmodule 

